QED cascades induced by circularly polarized laser fields

N. V. Elkina,A. M. Fedotov, I. Yu. Kostyukov, M. V. Legkov, N. B. Narozhny, E. N. Nerush, and H. Ruhl, PHYSICAL REVIEW SPECIAL TOPICS - ACCELERATORS AND BEAMS 14, 054401 (2011)
Notebook: Óscar Amaro, November 2022 @ GoLP-EPP

Introduction

In this notebook we reproduce some results from the paper.

In[2126]:=
```
(* Airy function *)
Clear[ξ, x]
Refine[
 (((1/π) Integrate[Cos[ξ^3 / 3 + ξ x], {ξ, 0, ∞}] // Normal) - AiryAi[x]) // Simplify,
 {x > 0}]
```

Out[2127]= 0

Equations 7, 8, 9

In[2128]:=
```
Clear[p0, p0x, p0y, 𝔼, 𝔼x, 𝔼y, E0, ω, sols, a0, m, c, e, t0, px, py, eq, χe]

(* eq 6 𝔼 field components *)
𝔼 = {E0 Cos[ω t], E0 Sin[ω t]};

(* eq 7 equation of motion *)
eq = {D[px[t], t], D[py[t], t]} == e 𝔼;

(* eq 8 solution *)
sols = DSolve[{eq, {px[t0], py[t0]} == {p0x, p0y}}, {px[t], py[t]}, t] // Simplify;
E0 = m ω c a0 / e;
sols = sols // FullSimplify;

(* t0=0 and initially at rest p0=0*)
px = sols[[1, 1, 2]] /. {t0 → 0, p0x → 0}
py = sols[[1, 2, 2]] /. {t0 → 0, p0y → 0}

(* get energy ϵe *)
ϵe = Refine[
   m c^2 Sqrt[(px / (m c))^2 + (py / (m c))^2 + 1] // Simplify, {m > 0, c > 0}] // Simplify
(* check eq 9a *)
ϵe - m c^2 Sqrt[1 + 4 a0^2 Sin[ω t / 2]^2] // Simplify

(* get energy χe *)
p = {px, py};
χe =
  Refine[(e ℏ)/(m^3 c^4) Sqrt[Refine[(Norm[(ϵe 𝔼)/c]^2 - (p.𝔼)^2) // Simplify, {a0 > 0, m > 0, c > 0,
       e > 0, ω > 0, t > 0}] // FullSimplify], {a0 > 0, m > 0, c > 0, e > 0, ω > 0, t > 0}]

(* check eq 9b *)
χe - (e ℏ E0)/(m^2 c^3) Sqrt[1 + 4 a0^2 Sin[ω t / 2]^4] // Simplify
```

Out[2134]= $a0\, c\, m\, \mathrm{Sin}[t\, \omega]$

Out[2135]= $a0\, c\, m\, (1 - \mathrm{Cos}[t\, \omega])$

Out[2136]= $c^2\, m\, \sqrt{1 + 2\, a0^2 - 2\, a0^2\, \mathrm{Cos}[t\, \omega]}$

Out[2137]= 0

Out[2139]= $$\frac{7.42462 \times 10^{-35}\, a0\, \omega\, \sqrt{2 + 3\, a0^2 + a0^2\, (-4\, \mathrm{Cos}[t\, \omega] + \mathrm{Cos}[2\, t\, \omega])}}{c^2\, m}$$

Out[2140]= $$-\frac{1.06911\times 10^{-50}\ \mathrm{a0}\ \omega\ \sqrt{2+3\ \mathrm{a0}^2-4\ \mathrm{a0}^2\ \mathrm{Cos}[t\ \omega]+\mathrm{a0}^2\ \mathrm{Cos}[2\ t\ \omega]}}{c^2\ m}$$

Equations 10, 11

In[2194]:=

```
Clear[p0, p0x, p0y, E, Ex, Ey, E0, ω,
 sols, a0, m, c, e, t0, px, py, eq, χe, r1, r2, ES, ℏ]

(* eq 10a *)
r1 = Refine[Asymptotic[m c^2 Sqrt[1 + 4 a0^2 Sin[ω t/2]^2], a0 → ∞],
  {a0 > 0, t > 0, ω > 0, Sin[t ω/2] > 0}]
Series[r1, {t, 0, 1}] /. {a0 → e E0/(m c ω)}

(* eq 10b *)
r2 = Refine[Asymptotic[(e ℏ E0)/(m^2 c^3) Sqrt[1 + 4 a0^2 Sin[ω t/2]^4], a0 → ∞],
  {a0 > 0, t > 0, ω > 0, Sin[t ω/2] > 0}]
(* check *)
r22 = Series[r2 , {t, 0, 2}] //. {a0 → e E0/(m c ω)} // Normal
(r22 - (1/2 (E0/ES)^2 (m c^2 ω)/ℏ t^2) // Simplify) /. {ℏ → (m^2 c^3)/(e ES)}

(* eq 11 t_acc: this is estimated without the √2 *)
tacc = Solve[(e^2 E0^2 t^2 ω ℏ)/(c^4 m^3) == 1, t][[2, 1, 2]]
(* check. α μ will not depend on α *)
Refine[(ℏ/(α m c^2 μ) Sqrt[(m c^2)/(ℏ ω)] - tacc) //. {μ → E0/(α ES), ES → (m^2 c^3)/(e ℏ)} // Simplify,
  {m > 0, c > 0, ω > 0, ℏ > 0}] // Simplify
```

Out[2195]= $2\, a0\, c^2\, m\, \mathrm{Sin}\left[\frac{t\,\omega}{2}\right]$

Out[2196]= $c\, e\, E0\, t + O[t]^2$

Out[2197]= $$\frac{2\, a0\, e\, E0\, \hbar\, \mathrm{Sin}\left[\frac{t\,\omega}{2}\right]^2}{c^3\, m^2}$$

Out[2198]= $$\frac{e^2\, E0^2\, t^2\, \omega\, \hbar}{2\, c^4\, m^3}$$

Out[2199]= 0

Out[2200]= $$\frac{c^2\, m^{3/2}}{e\, E0\, \sqrt{\omega}\, \sqrt{\hbar}}$$

Out[2201]= 0

Figure 4 a

In[2149]:=
```
Clear[χe, e, ℏ, E0, m, c, a0, ω, t, ES, χet2]

χe = (e ℏ E0)/(m^2 c^3) Sqrt[1 + 4 a0^2 Sin[(ω t)/2]^4];
χet2 = 1/2 (E0/ES)^2 * (m c^2 ω)/ℏ t^2; (*10b*)
E0 = (a0 m ω c)/e;

c = 3 × 10^8; (*[m/s]*)
e = 1.6 × 10^-19; (*[C]*)
ℏ = 1.05 × 10^-34; (*[Js]*)
m = 9.1 × 10^-31; (*[Kg]*)
ES = m^2 c^3 / (e ℏ); (*[V/m]*)

ω = e/ℏ; (*[s], 1eV *)
a0 = 2 × 10^4;

(* the red curve is only an approximation
 of the simulation data of the paper *)
Plot[{χe /. {t → tω/ω}, (χet2 HeavisideTheta[0.3 - tω] +
     (0.5 χet2 - 15) HeavisideTheta[tω - 0.3] HeavisideTheta[1.35 - tω] +
     (0.3 χet2 - 40) HeavisideTheta[tω - 1.35] HeavisideTheta[1.88 - tω] +
     (0.1 χet2 - 60) HeavisideTheta[tω - 1.88] HeavisideTheta[3 - tω]) /. {t → tω/ω}},
 {tω, 0, 2}, PlotRange → {0, 380}, PlotStyle →
  {{LightGray, Thickness[0.015]}, Red, Default},
 Frame → True, FrameLabel → {"t[ω^-1]", "χe"},
 PlotLegends → {"without QED effects", "''Fit''/Model of Run 1"}]
```

Out[2160]=



Figure 5

In[2161]:=
```
Clear[χe, e, ℏ, E0, m, c, a0, ω, t, ES, χet2]
```

$$\chi e = \frac{e\,\hbar\,E0}{m^2\,c^3}\,\mathrm{Sqrt}\left[1+4\,a0^2\,\mathrm{Sin}\left[\frac{\omega\,t}{2}\right]^4\right];$$

$$\chi et2 = \frac{1}{2}\left(\frac{E0}{ES}\right)^2 * \frac{m\,c^2\,\omega}{\hbar}\,t^2;\ (*10b*)$$

$$E0 = \frac{a0\,m\,\omega\,c}{e};$$

$c = 3\times10^8$; (*[m/s]*)

$e = 1.6\times10^{-19}$; (*[C]*)

$\hbar = 1.05\times10^{-34}$; (*[Js]*)

$m = 9.1\times10^{-31}$; (*[Kg]*)

$ES = m^2\,c^3 / (e\,\hbar)$; (*[V/m]*)

$\omega = 1\,\frac{e}{\hbar}$; (*[s], 1eV *)

$a0 = 5\times10^4$;

(* the red curve is only an approximation of the simulation data of the paper *)

$$\mathrm{Plot}\left[\left\{\left(\frac{c\,e\,E0\,t}{10^3\,m\,c^2}\,\mathrm{HeavisideTheta}[0.33-t\omega]+5\,\mathrm{HeavisideTheta}[t\omega-0.33]\right)/.\left\{t\to\frac{t\omega}{\omega}\right\}\right\},\right.$$

{tω, 0, 1.4}, PlotRange → {0, 18}, PlotStyle → {Red, Default}, Frame → True,

FrameLabel → {"t[ω^{-1}]", "ϵ[10^3mc^2]"}, PlotLegends → {"e$^-$ ''fit''"},

FrameTicks → {Automatic, {2, 4, 6, 8, 10, 12, 14, 16, 18}}]

Out[2172]=



Figure 7

In[2173]:=
```
Clear[χe, e, ℏ, E0, m, c, a0, ω, t, ES, Γ, μ, α]

c = 3 × 10^8; (*[m/s]*)
e = 1.6 × 10^-19; (*[C]*)
ℏ = 1.05 × 10^-34; (*[Js]*)
m = 9.1 × 10^-31; (*[Kg]*)
ES = m^2 c^3 / (e ℏ); (*[V/m]*)
α = 1 / 137; (*[]*)

(* eq 13 *)
Γ = α μ^(1/4) Sqrt[m c^2 ω / ℏ];

Plot[{Γ/ω /. {ω → 1 e/ℏ}, Γ/ω /. {ω → 0.66 e/ℏ}}, {μ, 0, 60},
 PlotRange → {0, 18}, Frame → True, FrameLabel → {"μ", "Γ/ω"},
 PlotStyle → {{Dashed, Darker[Green]}, {Dotted, Blue}},
 PlotLegends → {"estimation ℏω=1eV", "estimation ℏω=0.66eV"}]
```

Out[2181]=



Appendix

In[2182]:=
```
Clear[x]
D[AiryAi[x], {x, 2}] - x AiryAi[x]
-2/3 Integrate[x^3 AiryAi[x], {x, 0, ∞}]
```

Out[2183]= 0

Out[2184]= $-\frac{4}{9}$